

**MSX*****P-BASIC  
Reference  
Manual***

PERSONAL COMPUTER

**PX-7****PREFACE**

Thank you very much for having purchased the PX-7 Personal Computer manufactured by PIONEER.

The PX-7 is an MSX system developed using PIONEER's most advanced technology.

This manual is a guide to the original P-BASIC (PIONEER BASIC) language. P-BASIC is the extended BASIC that runs on the PX-7 as a part of MSX BASIC. The distinct functions of the PX-7, such as superimpose, system control, and stereo control can easily be controlled by the P-BASIC.

**KEEP THIS MANUAL IN A SAFE PLACE FOR FUTURE REFERENCE!!**

**CONTENTS**

FIRST THINGS .....	2
BLIND .....	9
DEF UNIV .....	11
EXTV .....	12
FRAME/CHAPTER .....	13
FRAME OFF/CHAPTER OFF .....	14
IMPOSE .....	15
LCOPY .....	16
LD .....	17
MUTE .....	18
PAN .....	19
REMOTE .....	20
SCLOAD .....	23
SCSAVE .....	24
SEARCH .....	25
SYMBOL .....	26
VIDEO .....	27

 **PIONEER**

# FIRST THINGS

---

## 1. Features of Extended P-BASIC

P-BASIC is the extended BASIC that runs on the PIONEER Personal Computer PX-7 as a part of MSX-BASIC. P-BASIC can be easily used to control the PX-7's superimpose, system control, and stereo control functions. P-BASIC adds the following features to MSX-BASIC.

- Instructions to control the superimpose function.
- Instructions to control the stereo effects and audio signals.
- Instructions to support the system control functions.
- Instructions to erase and redisplay the graphic elements in various sequences.
- Support of switching the picture and sound by the use of the SUPERIMPOSE, VIDEO and COMPUTER keys.

## 2. How to use the extended statements

To use the extended P-BASIC statements, place the keyword CALL before the statement to be used as in the following format:

CALL BLIND

or

\_\_BLIND (Use SHIFT + ⇐ to enter\_\_)

The statements can usually be used in either the direct mode or the program mode; however, some of them are only valid when the screen is in mode 2 (high resolution graphics mode). Those statements cannot be used in the direct mode, where screen mode 2 is not available.

**NOTE:**

*P-BASIC is available only when the '1' key is pressed when selecting the BASIC mode.*

### 3. How to interpret the P-BASIC reference section

#### 1. Extended statement:

A simple expression of statement (command) functions.

#### 2. Format:

Indicates the method of writing statements. When actually performing inputs, abide by the following rules:

- When inputting commands, either upper case or lower case characters may be used. However, character strings enclosed by quotation marks " " (file names, etc.), must be written while discriminating upper and lower case.
- The user may designate items enclosed in angle brackets < >.
- Items enclosed in square brackets [ ] are options, and may be eliminated if desired. When eliminated, the eliminated value (the value set the P-BASIC), or the previous value is applied.
- Some multiple parameters separated by commas , ; if they may be eliminated, they are written in a formula like that shown below:

```
CALL SYMBOL (X,Y), < chr> [, < hor> ], < ver> ], < col> ], < rot> ]
```

In this case, the parameters enclosed in the square brackets may be eliminated. Here, when all subsequent parameters are to be eliminated, they will be eliminated together with their commas, but when following parameters are to be designated, it is necessary to specify all preceding commas.

For instance, in the example above, when designated only < col>, the following pattern is used:

```
CALL SYMBOL (X,Y), < chr> ,, < col>
```

#### 3. Action:

Explanations of uses, detailed functions, and precautionary items with respect to commands.

#### 4. Example:

Indicates a program example (and results of execution) in which several statements have been changed.

## 4. Summary of P-BASIC statements

The P-BASIC statements can be generally divided into the four following types of functions:

—Superimpose control

EXTV  
IMPOSE  
VIDEO

—Stereo effects control

MUTE  
PAN

—System control

DEF UNIV  
FRAME/CHAPTER  
FRAME OFF/CHAPTER OFF  
LD  
REMOTE  
SEARCH

—Graphics control

BLIND  
LCOPY  
SCLOAD  
SCSAVE  
SYMBOL

## 5. Explanation of System Control

As an example of using the system control functions, the control of the LD-700 Laser vision player will be explained.

### 1. REMOTE

The REMOTE instruction outputs the system control codes. A maximum of 16 different devices may be controlled.

The functions specified are composed of two characters, as shown in the system control function code table. The plus sign (+) between the two characters may be omitted.

Depending on the device, there are functions that cannot be used; therefore, please read the functions in the table carefully.

The sample program below controls play, still, step forward, step reverse, and frame search of a player.

```

10 'PLAYER START
20 CALL REMOTE (0,"PWS")
30 'STILL 100 STEP FORWARD
40 FOR I = 1 TO 100
50 CALL REMOTE (0,"S")
60 NEXT I
70 'STILL 50 STEP REVERSE
80 FOR I = 1 TO 50
90 CALL REMOTE (0,"S - ")
100 NEXT I
110 'SEARCH 10000
120 CALL IMPOSE (0)
130 CALL SEARCH (0,F,10000)
140 CALL REMOTE (0,"WP")
150 CALL IMPOSE (2)
160 END

```

## 2. DEF UNIV

In the REMOTE command, devices 3 to 15 are unspecified. The DEF UNIV command is used to specify the allocation of the device.

```
CALL DEF UNIV (5,&HA0)
```

The above instruction assigns device 5 to the apparatus having &HA0 as a device code. After this statement is evoked, if a 5 is specified in a REMOTE statement, the control codes will be sent to the apparatus having &HA0 as a device code.

```

10 CALL DEF UNIV (5,&HA0)
20 CALL REMOTE (5,"P")
:
:
100 CALL REMOTE (5,"P/")
:
:

```

### 3. SEARCH

The SEARCH function is used with the LD-1100 and LD-700 Laser vision players, and allows the control of jumping playback to any frame or chapter. (There are also time search discs among CLV discs. The format for specifying a time search is the same as for a frame search.)

During the search action since the player cannot output video signals, it is best to place the system in the computer mode using the IMPOSE statement before using SEARCH. The end of the search is confirmed by the potential to output video signals. (Refer to the section on VIDEO WAIT.)

When the search is complete and CAV discs are used, the screen will be in still state, and therefore, to continue playback, specify the play function.

#### Example SEARCH program

```
10 CALL IMPOSE (0)
20 CALL SEARCH (0,F,10000)
30 CALL REMOTE (0,"W")
40 CALL IMPOSE (2)
50 CALL REMOTE (0,"P")
```

### 4. VIDEO WAIT

It takes several seconds for the picture to appear after the play button on a video disc player is pushed. This also occurs when the SEARCH command is used and the screen is blank during the interval of the search. P-BASIC has a function to detect this screen blank condition so that it does not show, and the user program can confirm the player being READY or the search being complete.

**CALL REMOTE (0,"W")**

The above command makes the system wait until a video signal is input. The "W" is the character used to specify the video wait function. This same function can also be accomplished with the EXTV command.

```
10 CALL EXTV (A)
20 IF A = 0 THEN 10
```

EXTV is a command used to detect the condition of external video signals; however, when it is used in the above manner, it can be used to create the video wait function.

**5. FRAME/CHAPTER**

The FRAME and CHAPTER commands are mainly for use with the LD-700 Laser vision player and cannot be used with the LD-1100. These commands execute the specified subroutine when the playback of the specified frame or chapter is detected.

Do not use other control commands when using the FRAME or CHAPTER commands. Once a FRAME or CHAPTER command is executed, if yet another control command is executed before the specified frame or chapter is played back, the confirmation of the frame or chapter will be incorrect. This will lead to incorrect execution of the subroutine.

In addition, once the subroutine is executed, the FRAME or CHAPTER function ends.

The FRAME OFF/CHAPTER OFF commands are used to stop the functions of the FRAME and CHAPTER commands; thus, after the FRAME OFF/CHAPTER OFF commands are executed and the specified FRAME or CHAPTER is played back, the subroutine is not executed.

**Bad example**

```
10 CALL REMOTE (0,"PW")
20 CALL FRAME (100,GOSUB100)
30 CALL REMOTE (0,"P")
:
100 .....
```

After the command in line 30 is executed, the subroutine beginning at line 100 is called at the wrong time.

**6. SPECIFYING CONTINUOUS ACTION, SCAN, FAST ON THE LD-1100 and LD-700**

When specifying continuous-action, scan, fast multi-speed, and slow on the LD-1100 and LD-700 Laser vision players, a period is placed in front of the character specifying the function. If a period is not specified, the action specified will be done intermittently.

**NOTE:**

*Period option is using with LD-700, rarely an "Illegal function call" error message is printed. In this case, please use the DEF UNIV command for the LD-700 to change another device No. (3 to 15).*

**Example of continuous action (Period specified)**

```
10 CALL REMOTE (0,"PW")
20 CALL DEF UNIV (3,&HA8)
30 FOR I = 1 TO 100
40 CALL REMOTE (3,".F+ ")
50 NEXT I
```

**Example of intermittent action (No period specified)**

```
10 CALL REMOTE (0,"PW")
20 CALL DEF UNIV (3,&HA8)
30 FOR I = 1 to 100
40 CALL REMOTE (3,"T+ ")
50 NEXT I
```



**Extended statement**

Erases or redisplay the graphic screen.

**Format:** CALL BLIND ([<expression>],  $\left| \begin{array}{c} S \\ L \end{array} \right|$ )

**Action:** On the value specified in the expression, the graphic screen is erased or redisplayed in the various sequences.

<expression> The values from 0 to 9 specify the sequence for erasure or redisplay of the graphic screen. If a value is omitted, the erasure or redisplay of the graphic screen is done instantaneously without regard to the sequence.

[S/L] Specifies the method of handling the graphic data. When S is specified, the graphic data is saved as it is. At the same time, the graphic screen is erased in the given sequence. When L is specified, the graphic data which was saved by specifying S in a previous command is loaded and the graphic screen is redisplayed. When neither S nor L are specified, the screen is erased in the specified sequence.

The sequences for the value specified in <expression> are as follows:

<expression> = 0



n=0

Top to bottom

<expression> = 1



n=1

Bottom to top

<expression> = 2



n=2

From the center to the top  
and bottom edges

<expression> = 3



n=3

From the top and bottom  
edges to the center

# BLIND

<expression> = 4



Top to bottom as in a venetian blind

<expression> = 5



Left to right as if drawing a curtain

<expression> = 6



From the center out

<expression> = 7



From the outside to the center

<expression> = 8



From the middle to the left and right

<expression> = 9



From the left and right to the middle

## NOTE:

The *BLIND* command is only available in the *SCREEN 2* mode.

## Example:

```
10 SCREEN 2
20 FOR X = 0 TO 255 STEP 16
30 LINE (X,0) - (X+15,191), X/16, BF
40 NEXT X
50 CALL BLIND (,S)
60 FOR K = 0 TO 9
70 CALL BLIND (K,L)
80 CALL BLIND (K)
90 NEXT K
100 END
```

## DEF UNIV

---

### Extended statement

Define the device to be controlled by the REMOTE statement.

**Format:**       CALL DEF UNIV (<expression 1>, <expression 2>)

**Action:** The DEF UNIV command defines the device to correspond to system control. The device number, from 3 to 15, to be used in the REMOTE command is placed in <expression 1>, and the device code the will correspond to the device number is placed in <expression 2>.

Valid values for <expression 1> are between 3 and 15.

Valid values for <expression 2> are between 1 and 255 (&HFF)

# EXTV

---

## Extended statement

Used to switch the external video signal on or off.

**Format:** CALL EXTV (variable)

**Action:** The EXTV command is used to verify if there is a video input signal at the video input terminal and returns a value according to the result. The value, 0 or 1, has the following meaning.

- 0: An external video signal is not being input.
- 1: An external video signal is being input.

## FRAME / CHAPTER

---

### Extended statement

Specifies the first line of the subroutine that is executed at the arrival at the specified FRAME/CHAPTER of the Laser vision disc.

**Format:**       CALL FRAME (<expression>, GOSUB <line number>)  
                  CALL CHAPTER (<expression>, GOSUB <line number>)

**Action:** The FRAME and CHAPTER commands are for use with the PIONEER LD-700 Laser vision player. When the frame or chapter specified in <expression> is played back, the subroutine beginning at <line number> is executed. CALL FRAME and CALL CHAPTER cannot be used simultaneously. Once the specified subroutine is executed, the action of this command ends.

This command is only valid when the player is being played back, and it is invalid if the player is stopped or is doing a search. Valid values for the <expression> are between 50 and 54000.

### Example:

```
10 CALL REMOTE (0,"P+W")
20 CALL IMPOSE (2)
30 CALL FRAME (100,GOSUB 50)
40 GOTO 40
50 CALL REMOTE (0,"S")
60 RETURN
```

When FRAME 100 of the disk is played back, line 50 is executed.

# FRAME OFF/CHAPTER OFF

---

## Extended statement

Cancels the CALL CHAPTER and CALL FRAME commands.

**Format:**           CALL |    FRAME    | OFF  
                      |    CHAPTER  |

**Action:** The CALL FRAME OFF and CALL CHAPTER OFF commands cancel the functions of the CALL FRAME (N, GOSUB In) and CALL CHAPTER (N, GOSUB In) commands; therefore, if they are executed before the subroutine specified in the latter two commands, even if the condition is met, the subroutine is not called.

## Example:

```
10 CALL REMOTE (0,"P+W")
20 CALL IMPOSE (2)
30 CALL FRAME (100,GOSUB 60)
40 CALL FRAME OFF
50 GOTO 50
60 CALL REMOTE (0,"S")
70 RETURN
```

This example adds FRAME OFF to the program used to illustrate FRAME (N, GOSUB In), except that since in line 40 that function is cancelled, even though FRAME 100 is played back, the subroutine is not executed.

# IMPOSE

## Extended statement

Switches the superimpose mode on the display.

**Format:** CALL IMPOSE (<expression>)

**Action:** The IMPOSE command is used to switch the superimpose mode. Valid values for <expression> are between 0 and 2; the relationship between the values and the modes are as follows.

Value of <expression> specified mode	Function	Background and Border colors
0: Computer mode	Only the computer screen is displayed	No change
1: Composite mode (superimpose mode)	Both the computer screen and the video input screen are displayed	COLOR „0 is executed and the border color becomes transparent (except SCREEN 0; SCREEN 0 has no border)
2: External video mode	Only the video input screen is displayed	COLOR „0 is executed and the background and border colors become transparent

Note that when there is no external video signal being input, even if 1 or 2 is specified, there is no change.

## Example:

```
10 ON KEY GOSUB 110,120,130
20 KEY (1) ON: KEY (2) ON: KEY (3) ON
30 SCREEN 2: COLOR „0,0: CLS
40 CIRCLE (128,96),90,4
50 PAINT (128,96),4
60 GOTO 60
110 CALL IMPOSE (0): RETURN
120 CALL IMPOSE (1): RETURN
130 CALL IMPOSE (2): RETURN
```

The screen mode is switched using programmable function keys F1, F2, and F3.

## **LCOPY** (Available in SCREEN 2 mode)

---

### **Extended statement**

Outputs a hard copy of the display screen.

**Format:**       CALL LCOPY [(**<expression>**)]

**Action:** The SCREEN 2 mode screen is copied on the printer according to the value in **<expression>**. If the value in **<expression>** is 0 or 1, the output is done as follows.

0: A positive copy of the screen is made

1: A reverse copy of the screen is made

If the value in **<expression>** is omitted, the default is 0.

When there are multiple foreground colors due to the use of the PSET and PAINT commands, the output may not be exactly as expected. In that case, limit the colors used for the foreground and background to two colors, one for the foreground and one for the background.



## Extended statement

Executes the interactive software from a CPE disc.

**Format:** After the computer program encoded disc (CPE DISC) is set on the Laser vision player, if CALL LD is input, the program on the disc is automatically read and the program thus read is executed.

# MUTE

---

## Extended statement

The stereo audio input signal is muted (cancelled).

Format:

$$\text{CALL MUTE} \left[ \begin{array}{c} \text{R} \\ \text{L} \\ \text{OFF} \end{array} \right]$$

**Action:** The MUTE command is used to mute (cancel) the stereo audio input signals and can be used with one channel (right or left) at a time.

The right and left channels are specified as shown below.

Command	Function	Stereo input signal	
		R	L
MUTE	Mutes both the R and L channels	X	X
MUTE R	Mutes the R channel only	X	O
MUTE L	Mutes the L channel only	O	X
MUTE OFF	Cancels the muting function	O	O

\*X: Signal muted

O: Signal not muted

# PAN

---

## Extended statement

Outputs the sound from any specified position.

**Format:** CALL PAN (<X axis>, <volume>, <character string>)

**Action:** The sound generated by the music macro instruction is output at the position specified in the <X axis> parameter.

<X axis> Specify a value between 0 and 255. When 0 is specified, the sound is positioned at the extreme left, and when 255 is specified, the sound is positioned at the extreme right.

<volume> Specify a value between 0 and 7. When 0 is specified the volume is approximately that of V8 specified in the PLAY command, for 7 the volume is approximately that of V15.

<character string> Same as the music macro instruction in the PLAY command. Note that in the PAN command, V, S, M, and X cannot be specified in this character string and that the length of this character string cannot exceed 79 characters.

## Example:

```
10 FOR X=0 TO 255 STEP 7
20 CALL PAN (X,0,"C")
30 NEXT X
40 FOR X=255 TO 0 STEP -7
50 V=INT (X/32)
60 CALL PAN (X,V,"C")
70 NEXT X
80 END
```

Lines 10 to 30 pan the C note from the left to the right.

Lines 40 to 70 decrease the volume as the signal is panned from right to left.

# REMOTE

---

## Extended statement

Used to control external devices.

**Format:** CALL REMOTE (<expression>, <character string>)

**Action:** The REMOTE command is used to control external devices, and the control codes are sent to the device according to the character strings.

<expression> specifies the external device to be controlled, with valid values are between 0 and 15. The external devices corresponding to 0, 1, and 2 are already assigned as in the following. External devices using the values from 3 to 15 may be assigned using the CALL DEF UNIV command.

- 0: PIONEER Laser vision player LD-700
- 1: PIONEER Laser vision player LD-1100
- 2: PIONEER component display SD-26

<character string> Refer to the System Control Function Tables for meanings of characters specified. A maximum of 16 functions may be specified in the <character string>.

Note that the plus (+) mark in each function may be omitted. For example, REMOTE (0, "P+") is the same as REMOTE (0, "P").

## System Control Function Table

The functions available are those included with the remote control unit for each device.

## LD-700 description and functions

Description	Code	Function	Description	Code	Function
A+	48	Repeat A	M+	58	Multi-speed forward (▶)
A-	44	Repeat B	M-	55	Multi-speed reverse (◀)
C+	47	Multi-speed up	P+	17	Play (▶)
C-	46	Multi-speed down	P@	16	Reject
D+	43	Frame/chapter number display	P/	18	Pause
F+	10	Scan forward (▶▶)	S+	54	Still; step forward
F-	11	Scan reverse (◀◀)	S-	50	Still; step reverse
L+	4B	Audio 1/left	T+	51	Fast forward: three times normal speed (x▶)
L-	49	Audio 2/right	T-	59	Fast reverse: three times normal speed (x◀)
L@	4A	Stereo	X+	45	Clear

Note: See page 8.

## LD-1100 description and functions

Description	Function	Description	Function
D+	Frame number display	P+	Play (▶)
D-	Chapter number display	P@	Reject
F+	Scan forward (▶▶)	P/	Pause
F-	Scan reverse (◀◀)	S+	Still; step forward (▶ )
L+	Audio 1/left	S-	Still; step reverse ( ◀)
L-	Audio 2/right	T+	Fast forward: three times normal speed (x▶)
M+	Slow forward ( ▶)	T-	Fast reverse: three times normal speed (◀x)
M-	Slow reverse (◀ )		
Used when scan, 3-times normal speed or other function is continuous. Example: REMOTE (1, ".F+.F+.F+")			

Table of SD-26 Codes and Functions

Symbol	Code	Function	Symbol	Code	Function
1	01	Channel: A	F+	10	Channel up (+)
2	02	Channel: B	F-	11	Channel down (-)
3	03	Channel: C	K1	0C	Input selector: TV
4	04	Channel: D	K2	0D	Input selector: Video Disc
5	05	Channel: E	K3	0E	Input selector: Video 1
6	06	Channel: F	K4	0F	Input selector: Video 2
7	07	Channel: G	O@	1C	Power ON/OFF
8	08	Channel: H	V+	0A	Volume up (+)
9	09	Channel: I	V-	0B	Volume down (-)
0	00	Channel: J		48	Sleep
C-	46	Channel: K		49	Mute
C+	47	Channel: L		4A	Display Call
	4D	Channel: M			
	4E	Channel: N			
	4F	Channel: O			
	50	Channel: P			

Other functions

Description	Code	Function
M@	ID	Tape monitor on/off (for stereo amplifier)
P-	15	Direction/reverse play (for tape deck)
W		Video wait (for Laser vision player)
R+	14	Recording (for tape deck)
R-	12	Recording mute (for tape deck)

### Extended statement

Loads screen data from cassette tape into the Video RAM (VRAM) for display on the screen.

**Format:**       CALL SCLOAD [("**<file name>**")]

**Action:** The SCLOAD command is used to search for and load the file specified in **<file name>** from cassette tape onto the screen (Video RAM or VRAM). During the search, if a file name that does not match the specified file name is found, the message **Skip: <file name>** is displayed. If the specified file name is found, the message **Found: <file name>** is displayed and the screen data is loaded.

The **<file name>** must be entered in six characters or less. If seven or more characters are specified, the first six characters are taken as the entry. If the **<file name>** is omitted, the first file found on the cassette is loaded.

If SCLOAD is executed, the SCREEN mode is automatically changed and the screen data appears on the screen.

If the SCLOAD command is executed in the direct mode, the system is returned to the command level at the end of execution. Thus, when the data is completely loaded, the screen will be switched from SCREEN 2 to SCREEN 1 or SCREEN 0, and the data will be lost.

## SCSAVE (Available only in SCREEN 2 mode)

---

### Extended statement

Saves the screen data on a data recorder.

**Format:** CALL SCSAVE ("<file name>"[,<baud rate>])

**Action:** The SCSAVE command saves the screen data in binary format onto cassette tape.

The data thus saved by the SCSAVE command can later be read into the VRAM by a SCLOAD command.

The maximum length of the name used in the <file name> is six characters. If seven or more characters are specified, the first six characters are taken as the file name and the remaining characters are ignored.

The value specified in <baud rate> sets the data transfer rate according to the following chart. If the <baud rate> is omitted, the baud rate specified in the SCREEN command is used.

- 1: 1200 baud
- 2: 2400 baud

To abort the SCSAVE command, hold the **CTRL** key down while pressing the **STOP** key.



# SEARCH

---

## Extended statement

The SEARCH command searches the Laser vision player for the specified screen.

**Format:** CALL SEARCH (<expression 1>,  $\left| \begin{array}{c} F \\ C \end{array} \right|$ , <expression 2>)

**Action:** The SEARCH command executes a search for the specified frame, chapter, or time on a Laser vision player.

<expression 1> Specifies the type of Laser vision player. If 0 is specified, the LD-700 player is being used, and if 1 is specified, the LD-1100 player is being used.

(F/C) Specifies frame or chapter search

F: Frame search/time search

C: Chapter search

<expression 2> Specifies the frame or chapter number. This value must be between 0 and 54000.

For a time search, one digit is used for the hours, and two digits are used for the minutes.

### Example:

```
10 CALL REMOTE (0,"P+W")
20 CALL SEARCH (0,F,10000)
30 CALL EXTV (A): IF A=0 THEN 30
40 CALL IMPOSE (1)
50 CALL REMOTE (0,"P+")
```

The above sample program is for use with the LD-700 Laser vision player. When using the LD-1100 Laser vision player, substitute the number 0 in lines 10, 20, and 50 with 1.

The above program starts the player and begins playback from frame 10000.

## SYMBOL (Available in SCREEN 2 mode)

---

### Extended statement

The SYMBOL statement displays the specified character string in the specified expanded size.

**Format:** CALL SYMBOL (X,Y), <chr> [,<hor>][,<ver>][,<col>]  
[,<rot>]

**Action:** The X and Y axes specify the leftmost-uppermost point at which the character string is to be displayed, in the specified color code and rotation. This command can only be used in SCREEN 2 mode.

<hor>:: The horizontal multiplier for display of the character string on the screen must be between 1 and 32. When this value is omitted, 1 is assumed.

<ver>:: The vertical multiplier for display of the character string on the screen must be between 1 and 24. When this value is omitted, 1 is assumed.

<col>:: Specify a value between 0 and 15 as the color code in the same format as the COLOR statement. When this value is omitted, the default color used is the one specified in this COLOR statement.

<rot>:: Specify a value between 0 and 3 as in the following. When this value is omitted, 0 is assumed.

**A**

0: Normal

**➤**

1: 90 degrees clockwise

**∨**

2: 180 degrees clockwise

**⤵**

3: 270 degrees clockwise

**NOTE:**

To specify a graphic character in the SYMBOL instruction, use the format below.

CALL SUMBOL (X,Y), CHR\$(Graphic character code) [,<hor>]  
[,<ver>] [,<col>][,<rot>]

# VIDEO

---

## **Extended statement**

Return the value of the superimpose mode.

**Format:**       CALL VIDEO (variable)

**Action:** The VIDEO statement is used to determine the current video display mode and returns a value in the variable specified. The value returned in the variable is between 0 and 2, and has the following meaning.

- 0: Computer mode
- 1: Superimpose mode
- 2: External video mode